# $Id$
# Maintainer: Gaetan Bisson <bisson@archlinux.org>
# Contributor: Thomas Dziedzic <gostrc at gmail>
# Contributor: Juergen Hoetzel <juergen@archlinux.org>
# Contributor: Tobias Kieslich <tobias@justdreams.de>

pkgname=gts
pkgver=0.7.6
pkgrel=3
pkgdesc='Provides useful functions to deal with 3D surfaces meshed with interconnected triangles'
url='http://gts.sourceforge.net/'
license=('LGPL')
arch=('i686' 'x86_64')
depends=('glib2')
source=("https://downloads.sourceforge.net/project/${pkgname}/${pkgname}/${pkgver}/${pkgname}-${pkgver}.tar.gz")
sha256sums=('059c3e13e3e3b796d775ec9f96abdce8f2b3b5144df8514eda0cc12e13e8b81e')

build() {
	cd "${pkgname}-${pkgver}"
	export LDFLAGS=${LDFLAGS//--as-needed,}
	./configure --prefix=/usr
	make
}

package() {
	cd "${pkgname}-${pkgver}"
	make DESTDIR="${pkgdir}" install
}
